# Diagonal Pair Coulomb Density Matrix on a Sphere: a numerical study

Riccardo Fantoni*
*Università di Trieste, Dipartimento di Fisica, strada Costiera 11, 34151 Grignano (Trieste), Italy*
(Dated: June 17, 2026)

We tabulate, on the diagonal, the $n$th power of the primitive approximation for the pair Coulomb density matrix on a sphere, with up to $n = 5$, and we discuss its behavior at particles contact as $n$ grows. Our results show that even on a curved surface quantum statistical physics is fundamentally different from its classical counterpart. If in the latter one needs to artificially introduce a regularization of the divergent attractive Coulomb potential preventing particles contact, for example with a hard core, in the former this is not necessary.

Keywords: Pair Coulomb Density Matrix; Primitive Approximation; End-Point Approximation; Sphere; Cusp Condition

## CONTENTS



## I. INTRODUCTION

In a quantum many body (Monte Carlo) simulation of a Coulomb liquid it is often useful to have access to the low temperature pair Coulomb density matrix $\rho$. In the high temperature classical limit the *primitive approximation* $\rho_{\rm primitive}$ becomes exact. Within Path Integral Monte Carlo (PIMC) of a many body system [1] in order to access the quantum regime with less imaginary time discretization error it is important to use the *pair-product inter-action*. The imaginary time $t$ is related to the temperature $T$ of the liquid by $t = \beta = 1/k_BT$ with $k_B$ Boltzmann constant. The main ingredient to construct the pair-product inter-action is the inter-action of the pair density matrix or the *pair-inter-action*. Starting from $\rho_{\rm primitive}$ at a high temperature $T_0$ we can halve the temperature simply by squaring it [1] [2, 3] (see also section IV.G of Ref. [1]). This lower temperature pair density matrix already possesses different properties than the primitive approximation. In position representation, on the diagonal

$$\left|\frac{\phi(\boldsymbol{r}_1,\boldsymbol{r}_2)}{k_BT_0}\right| = \left|\ln\left[\frac{\rho_{\rm primitive}(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;1/k_BT_0)}{\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;1/k_BT_0)}\right]\right| \xrightarrow{\boldsymbol{r}_1\to\boldsymbol{r}_2} \infty, \tag{1.1}$$

---

* riccardo.fantoni@scuola.istruzione.it

[1] The squaring method can be used if an exact solution is known, so that successive squarings allow to reach temperatures $T_0/2^n$ for $n = 1, 2, 3, \ldots$.

where $\phi(\boldsymbol{r}_1, \boldsymbol{r}_2)$ is the pair Coulomb potential between particle 1 at $\boldsymbol{r}_1$ and particle 2 at $\boldsymbol{r}_2$. On the other hand, for a repulsive Coulomb potential, when Trotter formula holds [4], we can reach the temperature $T_0 = T_n/n$ sending $n \to \infty$ and $T_n \to \infty$, i.e. through the *continuum limit* [2]. In this case we find

$$|P(\boldsymbol{r}_1, \boldsymbol{r}_2; 1/k_B T_0)| = \lim_{n\to\infty} \left| \ln\left[\frac{[\rho_{\text{primitive}}]^n(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1, \boldsymbol{r}_2; n/k_B T_n)}{[\rho_0]^n(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1, \boldsymbol{r}_2; n/k_B T_n)}\right]\right| \xrightarrow{\boldsymbol{r}_1\to\boldsymbol{r}_2} \text{finite}, \tag{1.2}$$

where we denote with $[\rho]^n$ the $n$th power of the pair density matrix: $n = 2$ for the *square*, $n = 3$ for the *cube*, and so on. Moreover a *cusp condition* must hold for $P$ as the two particles approach each other, like in the flat space case [1, 5, 6].

The same contact value smearing property holds at any temperature $T_0$. This property makes it of fundamental importance not to use the primitive approximation for example in a simulation of an electron proton binary mixture where the divergence of the attractive Coulomb potential between unlike species at contact would inevitably let the fast electrons stick onto the slow protons, and consequently spoil the simulation. In other words quantum statistical physics does not need hard core regularization which is otherwise an essential but artificial feature in classical statistical physics necessary to avoid unlike particles collapse.

In this work we will show how this purely quantum smearing effect can be accomplished on a Sphere by approaching the $n \to \infty$ continuum limit. We will not give the exact solution for the pair Coulomb density matrix as was done in flat space in Refs. [6–9] but we will give an approximate numerical solution instead starting from the classical high temperature limit primitive approximation. If we were able to find an exact solution of the pair Coulomb density matrix on a sphere we would not need to worry about deriving it from the primitive approximation. In absence of an analytic solution here we propose to use the path integral discretized in imaginary time with timestep $\tau = 1/k_B T_n$ to reach a temperature $T_0 = 1/nk_B\tau$ approaching the continuum limit $\tau \to 0$. Of course this limit cannot be reached numerically but we will be able to say something about the trend as $n$ increases.

We will only worry about the *end-point approximation* (see section IV.F of Ref. [1]) to the pair Coulomb density matrix, i.e. the diagonal component of the pair-inter-action. Even if this is exact in the Euclidean three dimensional space due to the peculiar properties of the Coulomb pair potential [6] it becomes just an approximation on the Sphere.

## II. APPROACH TO THE CONTINUUM FOR THE PAIR COULOMB DENSITY MATRIX

We show a method to lower the temperature of the hot Pair Coulomb Density Matrix (PCDM) on a Sphere so to extend the work in flat (Euclidean) open (with open boundary conditions) space of Refs. [6, 8, 9] to a curved surface.

We will denote with a Greek index a coordinate component and with a Roman index a particle label. We will adopt Einstein summation convention to tacitly assume a sum over repeated Greek indexes. We distinguish between upper contravariant Greek indexes and lower covariant Greek indexes. Passing between the two is accomplished with a contraction with the metric tensor, as usual.

The metric tensor of the Sphere of radius $a$ is

$$||g_{\alpha\beta}|| = a^2 \begin{pmatrix} 1 & 0 \\ 0 & \sin^2\theta \end{pmatrix}. \tag{2.1}$$

with $g(\boldsymbol{r}) = \det||g_{\alpha\beta}|| = a^4 \sin^2\theta$. We use polar coordinates $\boldsymbol{r} = (r^1, r^2) = (\theta, \varphi)$ with $\theta \in [0, \pi]$ the polar angle and $\varphi \in [0, 2\pi[$, so that $\theta = 0$ is the north pole.

It is important to realize that on a curved surface the two body problem is generally not separable into a center of mass and a relative one body problem. Therefore, in full generality, we will study the position representation $\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t)$ of the density matrix $\rho = \exp(-\mathscr{H}t)$, where $\mathscr{H}$ is the Hamiltonian operator for the two bodies an $t$ is the imaginary time that in statistical physics is identified with the inverse temperature $\beta = 1/k_B T$ with $k_B$ Boltzmann constant. These are initially at $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$ and at an imaginary time $t$ later they are at $\boldsymbol{r}'_1$ and $\boldsymbol{r}'_2$.

Therefore, the PCDM on the Sphere will satisfy the Bloch equation

$$\frac{\partial\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t)}{\partial t} = \left(\frac{\lambda_1}{\sqrt{g(\boldsymbol{r}_1)}}\nabla_{1\alpha}\sqrt{g(\boldsymbol{r}_1)}\nabla_1^\alpha + \frac{\lambda_2}{\sqrt{g(\boldsymbol{r}_2)}}\nabla_{2\alpha}\sqrt{g(\boldsymbol{r}_2)}\nabla_2^\alpha - \phi(\boldsymbol{r}_1, \boldsymbol{r}_2)\right)\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; t), \tag{2.2}$$

$$\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}'_1, \boldsymbol{r}'_2; 0) = \delta(\boldsymbol{r}_1 - \boldsymbol{r}'_1)\delta(\boldsymbol{r}_2 - \boldsymbol{r}'_2)/\sqrt{g(\boldsymbol{r}_1)g(\boldsymbol{r}_2)}, \tag{2.3}$$

[2] For any finite $n$ one can treat the attractive Coulomb as well as the repulsive Coulomb.

where, for particle $i = 1, 2$, $\lambda_i = \hbar^2/2m_i$ with $m_i$ the mass of particle $i$, $g(\boldsymbol{r}_i) = a^4 \sin^2\theta_i$, $\boldsymbol{\nabla}_i = (\nabla_{i1}, \nabla_{i2}) = (\partial/\partial\theta_i, \partial/\partial\varphi_i)$ is the gradient respect to the coordinates $\boldsymbol{r}_i = (r_i^1, r_i^2) = (\theta_i, \varphi_i)$, $\delta$ is a two dimensional Dirac delta function, $\rho$ is the pair density matrix, and $\phi$ is the Coulomb pair potential.

The Coulomb pair potential for particles moving **on** the surface can be chosen as

$$\phi_{\rm on}(\boldsymbol{r}_1, \boldsymbol{r}_2) = Z_1 Z_2 e^2/d(\boldsymbol{r}_1, \boldsymbol{r}_2), \tag{2.4}$$

where $Z_i e$ is the charge of particle $i$ and for the distance between the two points $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$, $d(\boldsymbol{r}_1, \boldsymbol{r}_2)$, on the Sphere we can either choose the geodesic distance

$$d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) = a \arccos[\cos\theta_1 \cos\theta_2 + \sin\theta_1 \sin\theta_2 \cos(\varphi_1 - \varphi_2)], \tag{2.5}$$

or, viewing the Sphere as embedded in the three dimensional space, the Euclidean distance

$$d_e(\boldsymbol{r}_1, \boldsymbol{r}_2) = 2a \sin(d_g/2a). \tag{2.6}$$

For particles living **in** the surface [10] we may choose

$$\phi_{\rm in}(\boldsymbol{r}_1, \boldsymbol{r}_2) = -Z_1 Z_2 e^2 \ln[d_e(\boldsymbol{r}_1, \boldsymbol{r}_2)/\ell], \tag{2.7}$$

with $\ell$ a constant with the dimensions of a distance. We will call $\phi_{\rm on}^g$, $\phi_{\rm on}^e$, $\phi_{\rm in}$ the three cases just illustrated. The relationship between the intrinsic (scalar) $R$ and the extrinsic curvature $H$ of the Sphere of radius $a$ is $R = 2H/a = 2/a^2$.

We will measure energy in units of the Hartree $e^2/a_B$, with $e$ the charge of the electron and $a_B = \hbar^2/m_e e^2$ Bohr radius, with $m_e$ the mass of the electron. And length in units of the Bohr radius $a_B$. We will call $\mu_i = m_i/m_e$ the reduced mass of particle $i$. Temperatures are in units of $\hbar^2/m_e a_B^2 k_B$.

The free particles, $\phi \to 0$, pair density matrix can be determined from the work of Bastianelli and Corradini [11]. We will call their solution for the single free particle $i$, $\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i'; t)$. Therefore the pair free density matrix will be

$$\rho_0(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; t) = \rho_{10}(\boldsymbol{r}_1|\boldsymbol{r}_1'; t)\rho_{20}(\boldsymbol{r}_2|\boldsymbol{r}_2'; t). \tag{2.8}$$

Then our PCDM can be written as

$$\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; t) = \rho_0(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; t) e^{-P(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; t)}, \tag{2.9}$$

where $P$ is the *pair-inter-action* [1].

Aim of this work is to tabulate $P(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; t)$ for the three Coulomb cases illustrated above, $P_{\rm on}^g$, $P_{\rm on}^e$, $P_{\rm in}$. In the hot case, for small $t \ll 1$, for each of these cases, the *primitive approximation* [1]

$$P_{\rm primitive} \approx t\phi, \tag{2.10}$$

holds.

Therefore, introducing a timestep $\tau$, we can find the PCDM at all subsequent timesteps through the following convolution path integral

$$\begin{aligned} \rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; n\tau) = \int & \rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_{1,1}, \boldsymbol{r}_{2,1}; \tau)\rho(\boldsymbol{r}_{1,1}, \boldsymbol{r}_{2,1}|\boldsymbol{r}_{1,2}, \boldsymbol{r}_{2,2}; \tau) \dots \rho(\boldsymbol{r}_{1,n-1}, \boldsymbol{r}_{2,n-1}|\boldsymbol{r}_1', \boldsymbol{r}_2'; \tau) \times \\ & \prod_{k=1}^{n-1} \sqrt{g(\boldsymbol{r}_{1,k})g(\boldsymbol{r}_{2,k})} d\boldsymbol{r}_{1,k} d\boldsymbol{r}_{2,k}, \end{aligned} \tag{2.11}$$

where we denote with $\boldsymbol{r}_{i,k}$ the position of particle $i$ at timeslice $k$, $d\boldsymbol{r} = dr^1 dr^2 = d\theta d\varphi$, and we choose $\tau$ small enough so that

$$\rho(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; \tau) \approx \rho_0(\boldsymbol{r}_1, \boldsymbol{r}_2|\boldsymbol{r}_1', \boldsymbol{r}_2'; \tau) e^{-\tau[\phi(\boldsymbol{r}_1, \boldsymbol{r}_2) + \phi(\boldsymbol{r}_1', \boldsymbol{r}_2')]/2} \tag{2.12}$$

$$= \rho_{10}(\boldsymbol{r}_1|\boldsymbol{r}_1'; \tau)\rho_{20}(\boldsymbol{r}_2|\boldsymbol{r}_2'; \tau) e^{-\tau[\phi(\boldsymbol{r}_1, \boldsymbol{r}_2) + \phi(\boldsymbol{r}_1', \boldsymbol{r}_2')]/2}, \tag{2.13}$$

is a good approximation. We will perform the $4(n-1)$ integrations in Eq. (2.11) through a simple Monte Carlo (MC) quadrature [12] where a MC point is the tuple $(\boldsymbol{r}_{1,1}, \boldsymbol{r}_{2,1}, \dots, \boldsymbol{r}_{1,n-1}, \boldsymbol{r}_{2,n-1})$. It is important to stress that this is only possible if $Z_1 Z_2 > 0$ otherwise the Coulomb potential will be unbounded from below and the path integral of Eq. (2.11) breaks down [4].

According to the analysis of Ref. [11] we can further approximate

$$g^{1/4}(\boldsymbol{r}_i)\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i'; \tau) g^{1/4}(\boldsymbol{r}_i') \approx \frac{\mu_i}{2\pi\tau} \exp\left\{-\frac{\mu_i}{2\tau}\delta_{\alpha\beta}(\boldsymbol{x}_i - \boldsymbol{x}_i')^\alpha (\boldsymbol{x}_i - \boldsymbol{x}_i')^\beta - \tau\frac{V_R(x_i) + V_R(x_i')}{2}\right\}, \tag{2.14}$$

where $\delta_{\alpha\beta}$ is the Kronecker delta, the normalization $2\pi\tau/\mu_i$ is just the one of a free particle of mass $\mu_i$ on a plane and corresponds to the exact path integral performed with an action $\int_0^\tau dt'\,\mu_i\delta_{\alpha\beta}\dot{x}_i^\alpha\dot{x}_i^\beta/2$. Here $\boldsymbol{x}=(x^1,x^2)$ are the Riemann normal coordinates which are related to the polar coordinates $\boldsymbol{r}=(r^1,r^2)=(\theta,\varphi)$ through

$$x^1 = ar^1\cos r^2, \tag{2.15}$$

$$x^2 = ar^1\sin r^2, \tag{2.16}$$

$x=\sqrt{\delta_{\alpha\beta}x^\alpha x^\beta}$, and $V_R$ is the effective potential due to the scalar curvature of the Sphere, $R=2M^2$ with $M=1/a$, namely

$$V_R(x) = -\frac{M^2}{6} - \frac{5(Mx)^2-3+[(Mx)^2+3]\cos(2Mx)}{48x^2\sin^2(Mx)}, \tag{2.17}$$

which is equation (2.32) of Ref. [11]. [3] This effective potential diverges to minus infinity for $Mx=\theta=\pi$. This divergence will produce an exponential divergence in Eq. (2.14) which can be numerically prevented through a cutoff. This procedure should not alter significantly the main picture since it amounts to flattening a neighbor of the south pole of the sphere.

We may additionally introduce the following exact Feynman-Kac expression for the one free body density matrix at time $t$

$$g^{1/4}(\boldsymbol{r}_i)\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i';t)g^{1/4}(\boldsymbol{r}_i') = \left\langle e^{-S_R}\right\rangle, \quad i=1,2 \tag{2.18}$$

$$S_R = \int_0^t dt'\,V_R(x_i) \tag{2.19}$$

where $\langle\ldots\rangle$ denotes an average over all single particle Gaussian random walks from $\boldsymbol{r}_i$ to $\boldsymbol{r}_i'$ in a time $t$. In Appendix A we give the diagonal part $\rho_{i0}(\boldsymbol{r}_i|\boldsymbol{r}_i;t)=\rho_{i0}(\boldsymbol{0}|\boldsymbol{0};t)$ up to orders $t^8$. Our Eq. (A1) coincides with the perturbative expansion (3.23) of Ref. [11] for $d=2$ spatial dimensions.

Even if in general we will be interested in the full $P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1',\boldsymbol{r}_2';t)$ in practice in a many body simulation we may use the *end point approximation* for the inter-action as shown in section IV.F of Ref. [1] where one only needs the diagonal components $P(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t)=P(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)$. Where on a sphere the geodesic distance $d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)$ between any two points $\boldsymbol{r}_1$ and $\boldsymbol{r}_2$ is bounded from above by $a\pi$. Due to the Sphere symmetry, without loss of generality, we may choose $\boldsymbol{r}_1=(0,0)=\boldsymbol{0}$ and $\boldsymbol{r}_2=(\theta,0)$, so that $d_g=a\theta$.

On the diagonal we will have

$$\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t)=\rho(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)=\rho_0(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)e^{-P(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t)}, \tag{2.20}$$

we will then tabulate

$$P(m\delta;n\tau) = -\ln\left[\frac{\rho(m\delta;n\tau)}{\rho_0(m\delta;n\tau)}\right], \quad n=1,2,3,\ldots, \quad m=1,2,3\ldots,m_{\max} \tag{2.21}$$

for given small $\tau$ and $\delta$ so that $m_{\max}=\pi/\delta$. For an imaginary time $t_n=n\tau$ this will require a number of $4(n-1)$ quadratures for each $m$ to calculate $\rho(m\delta;n\tau)=\rho(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t_n)=\rho(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t_n)$ and $\rho_0(m\delta;n\tau)=\rho_0(d_g(\boldsymbol{r}_1,\boldsymbol{r}_2)/a;t_n)=\rho_0(\boldsymbol{r}_1,\boldsymbol{r}_2|\boldsymbol{r}_1,\boldsymbol{r}_2;t_n)$ for $m=1,2,3,\ldots,m_{\max}$, through the corresponding path integral convolution (2.11). Each convolution requires 4 quadratures over the azimuthal angle $\varphi\in[0,2\pi[$ and the polar angle $\theta\in[0,\pi]$ of each particle. In any case, in a many body simulation the pair-product inter-action is needed at a fixed small $t$, which determines the discretization error [1]. And one needs to tabulate the PCDM just once, at the beginning of the simulations.

In Fig. 1 we plot the ratio $R(m\delta;t)=P(m\delta;t)/t\phi(m\delta)$ as a function of $m$ for $a=1,\mu_1=\mu_2=1,Z_1=Z_2=1$, $t=n\tau=0.2,\delta=0.03$ and $n=2,3,4,5$. We see how at large particles distance the ratio $R$ approaches 1 but at contact it tends to a value below 1, approximately $\approx 1/n$. In between, $R_{\rm on}^g\approx R_{\rm on}^e$, have a peak at $d_g=\bar{d}_g(n)$ which moves at larger $d_g$ and grows in height as $n$ increases. The peak is at $(\bar{d}_g(n),\bar{R}_{\rm on}^g(n))$ where both $\bar{d}_g(n)$ and $\bar{R}_{\rm on}^g(n)$ depend on int$[(n-1)/4]$. The bottom panel of the figure shows the evolution of the peak at different values of the Sphere radius. It is important to note that the peak is an effect of the curvature of the surface and it disappears on a periodic flat square where the ratio is a monotonous increasing function of the distance reaching $\approx 1$ at the edge of the square box. This also shows the difference between a Sphere of infinite radius and a periodic flat square.

---

[3] This idea to map the fluid on the curved manifold with an equivalent one on Euclidean space but subject to an effective external potential due to the effect of the curvature has been used before even in the classical limit [13].




FIG. 1. We show the plot of the ratio $R(d_g/a;t) = P(m\delta;t)/t\phi(m\delta)$ as a function of $d_g = m\delta$ for $a = 1, \mu_1 = \mu_2 = 1, Z_1 = Z_2 = 1$, $t = n\tau = 0.2, \delta = 0.03$. On the top panel $R^g_{\rm on}$ for $n = 2, 3, 4, 5$. On the central panel the three Coulomb cases $R^g_{\rm on}, R^e_{\rm on}, R_{\rm in}$ for $n = 2$. On the bottom panel the case $R^g_{\rm on}$ for $n = 2$ and different Sphere radii. The angular convolutions in Eq. (2.11) were calculated with a Monte Carlo quadrature with no less than $10^6$ MC points and keeping the relative statistical error below 10% (this required approximately $> 10^{10}$ MC points in the $n > 4$ cases).

The behavior of the peak as a function of $n$ in the top panel, suggests that in the continuum limit $n \to \infty$ the peak will reach $d_g = \pi a$ and the ratio will be monotonously increasing throughout the whole Sphere. Differently from the periodic flat square, our results cannot exclude that on the Sphere the ratio can reach $R(\pi; t) > 1$ in the continuum limit.

In Table I we show $P(0; t)$ and $R(0; t)$ for $a = 1, \mu_1 = \mu_2 = 1, Z_1 = Z_2 = 1$, $t = n\tau = 0.2$ and increasing $n$. The value at contact, $d_g = 0$, is obtained through a linear extrapolation of the first two points at $m = 1, 2$ of the top panel of Fig. 1. We see how $R(0; t)$ decreases approximately as $1/n$ at increasing $n$. Therefore we expect that at constant $t = n\tau$ as $n \to \infty$, in the $\tau \to 0$ continuum limit, the ratio $R(0; t) \to 0$. Then we can say that $P(d_g \to 0; t)$ must be an infinite of order less than the infinite of $\phi(d_g \to 0)$. Moreover the PCDM at particles contact must obey to a cusp condition [6] as discussed in the next section.

TABLE I. Table for $P^g_{\rm on}(0; t)$ and $R^g_{\rm on}(0; t) = \lim_{\delta \to 0} P^g_{\rm on}(\delta; t)/t\phi^g_{\rm on}(\delta)$ for $a = 1, \mu_1 = \mu_2 = 1, Z_1 = Z_2 = 1$, $t = n\tau = 0.2$ and increasing $n$. The value at contact, $d_g = 0$, is obtained through a linear extrapolation of the first two points at $m = 1, 2$ of the top panel of Fig. 1.

| $n$ | $\tau$ | $P^g_{\rm on}(0; 0.2)$ | $R^g_{\rm on}(0; 0.2)$ |
|---|---|---|---|
| 2 | 0.1000 | 5.50 | 0.517 |
| 3 | 0.0666 | 4.31 | 0.364 |
| 4 | 0.0500 | 3.50 | 0.250 |
| 5 | 0.0400 | 3.10 | 0.207 |
| 6 | 0.0333 | 3.40 | 0.162 |

In Appendix B we present the FORTRAN code used to generate Fig. 1.

## III. CUSP CONDITION

On the sphere the cusp condition [1, 5, 6]. can be determined studying the asymptotic behavior as $\boldsymbol{r}_1 \to \boldsymbol{r}_2$ of a vanishing *residual energy* (defined in equation (4.6) of Ref. [1]) for the pair-product approximation to the inter-action. On a Sphere as $\boldsymbol{r}_1 \to \boldsymbol{r}_2$ we should compare

$$\left(\frac{\lambda_1}{\sqrt{g(\boldsymbol{r}_1)}}\nabla_{1\alpha}\sqrt{g(\boldsymbol{r}_1)}\nabla_1^\alpha + \frac{\lambda_2}{\sqrt{g(\boldsymbol{r}_2)}}\nabla_{2\alpha}\sqrt{g(\boldsymbol{r}_2)}\nabla_2^\alpha\right) P(d_g(\boldsymbol{r}_1, \boldsymbol{r}_2)/a; t) = -\phi(\boldsymbol{r}_1, \boldsymbol{r}_2). \tag{3.1}$$

As $\boldsymbol{r}_1 \to \boldsymbol{r}_2$, $d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) \to 0$ and we are looking at a local property on the sphere so that we can use the result on the tangent plane at the point $\boldsymbol{r}_1 = \boldsymbol{r}_2$ namely, for any $t$,

$$\lim_{d_g \to 0} \frac{\partial P_{\rm on}(d_g/a; t)}{\partial d_g} = \frac{Z_1 Z_2}{\lambda_1 + \lambda_2}, \tag{3.2}$$

$$\lim_{d_g \to 0} \frac{\partial P_{\rm in}(d_g/a; t)}{\partial d_g} = 0, \tag{3.3}$$

where we used the following expression for the Laplacian in an Euclidean space of dimension $d$ in an isotropic state

$$\boldsymbol{\nabla}_r^2 = \frac{\partial^2}{\partial r^2} + \frac{d-1}{r}\frac{\partial}{\partial r}. \tag{3.4}$$

## IV. CONCLUSIONS

We tabulated the (end-point approximation) diagonal inter-action for the repulsive $Z_1 Z_2 > 0$ pair Coulomb density matrix on the Sphere at a fixed temperature $T_0 = T_n/n$ obtained from the classical high temperature primitive approximation through a discretized path integral calculation with imaginary timestep $\tau = 1/T_n \to 0$ in the continuum $n \to \infty$ limit. We do this from a path integral calculated through a Monte Carlo quadrature. This only allows the determination of the repulsive ($Z_1 Z_2 > 0$) pair-inter-action because in the attractive ($Z_1 Z_2 < 0$) case the primitive pair-inter-action, i.e. the bare Coulomb pair-potential, is unbounded from below and Trotter formula breaks down. We took care explicitly of the $n = 2$ *square*, the $n = 3$ *cube*, up to $n = 5$ case. Higher $n$ cases become progressively more and more demanding computationally. We found that the ratio between the diagonal inter-action and its primitive approximation tends to $\approx 1/n$ at particles contact. This suggests that it will go to zero in the continuum limit $n \to \infty$ for a fixed temperature. If, for the case of two particles on a periodic square on a plane, the ratio have a monotonously increasing behavior from 0 to 1 for all $n$, on the Sphere the ratio has a monotonous behavior only in the continuum limit and it can reach a value bigger than 1 at the maximum geodesic distance between the two particles.

We also derive the cusp condition for the pair Coulomb density matrix on the Sphere. This has the same form as in flat space since as the two particles comes together at a point $\boldsymbol{r}$ their diagonal pair Coulomb density matrix reduces to the one on the tangent plane at $\boldsymbol{r}$.

For any finite $n$ one can treat the attractive $Z_1 Z_2 < 0$ Coulomb component provided some caution is taken numerically (see Appendix B). This amounts to introduce a numerical cutoff, imposing $d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) > \sigma$, or, in physical terms, a hard core between the two unlike particles. For the repulsive component we do not have this problem. We showed, numerically on a Sphere, the progressive smearing property of the divergence of the pair-inter-action of the primitive approximation obtained taking the $n$th powers of the primitive approximation for increasing $n$. Our results confirm how quantum statistical physics is fundamentally different from classical statistical physics: if in the latter one needs to artificially introduce a regularization of the divergent Coulomb potential preventing particles contact, for example with a hard core, in the former this is not necessary since the divergencies in the Coulomb pair-inter-action are washed out.

If one is looking for the exact attractive $Z_1 Z_2 < 0$ component of the pair Coulomb density matrix it is not possible to derive it through a path integral from the primitive approximation although one can gain numerically some insight by progressively reducing the extent $\sigma$ of the hard core and increasing the number $n$ of timeslices.

Even we are still far from an exact analytic solution of the Hydrogen atom problem on the surface of a Sphere with the path integral method, akin to what was done in Euclidean space in Ref. [7], nonetheless we found some expected numerical results.

## Appendix A: Diagonal free particle density matrix

Calling $\mu_i = m_i/m_e$ and $R = 2/a^2$ the Sphere scalar curvature, we find from equation (3.23) of Ref. [11]

$$\begin{aligned}\ln\left[2\pi t\sqrt{g(\mathbf{0})}\rho_{i0}(\mathbf{0}|\mathbf{0};t)/\mu_i\right] = {} & \frac{tR}{12} + \\ & \frac{(tR)^2}{6!}\frac{1}{2} + \\ & \frac{(tR)^3}{9!}16 + \\ & \frac{(tR)^4}{10!}\frac{59}{4} + \\ & \frac{(tR)^5}{11!}\frac{58}{3} + \\ & \frac{(tR)^6}{13!}\frac{550789}{1260} + \\ & \frac{(tR)^7}{14!}\frac{46838}{45} + \\ & \frac{(tR)^8}{17!}\frac{596004707}{720} + \mathscr{O}(t^9),\end{aligned} \tag{A1}$$

where due to the Sphere isotropy $\sqrt{g(\mathbf{0})}\rho_{i0}(\mathbf{0}|\mathbf{0};t) = \sqrt{g(\boldsymbol{r})}\rho_{i0}(\boldsymbol{r}|\boldsymbol{r};t)$ for any $\boldsymbol{r}$.

## Appendix B: The code

We give here the FORTRAN code listing to generate the table of Eq. (2.21). To tabulate the unlike attractive $Z_1 = -Z_2$ case it is necessary to choose `epsd` not too small, say 0.1 in order to prevent the overflow of the exponential of the attractive Coulomb potential when $d_g(\boldsymbol{r}_1, \boldsymbol{r}_2) <$`epsd`.

```
      program PCDMS
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
c
c     Pair Coulomb Density Matrix for End-Point Approximation
c     on a Sphere from the Primitive Approximation
c
c     Monte Carlo (MC) quadrature with relative error epsmc
c
c     0<ph<2*pi   0<th<pi   th=0 north pole
c     (ds/a)**2 = dph**2 + [sin(th)*dth]**2
c
c     a     = Sphere radius
c     tau   = time step
c     delta = geodesic distance step
c
cccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccccc
      implicit none
      integer*8 mni
      real*8    pi,a,delta,tau,epsmc,epsd
      parameter(mni=10)
      parameter(a=1.,delta=0.03,epsmc=1.d-1,epsd=1.d-6)
      real*8    mu1,mu2,z1,z2,ti,dth,dph
      real*8    th10,ph10,th20,ph20,dg0,de0,ccc
      real*8    th1,ph1,th2,ph2,dg,de
      real*8    th1p,ph1p,th2p,ph2p,dgp,dep
      real*8    rhoid,rho,pot,pcdm,fa,pidm,faid,pcdm2,pidm2
      real*8    rhor,rhoidr
      real*8    sigmac,sigmai,avpcdm,avpidm,avpcdm2,avpidm2
      real*8    ith1(mni),iph1(mni),ith2(mni),iph2(mni)
      integer*8 i,j,k,ni,id,nd,nmc,mnmc
      character potk*3,tabf*20

      pi=acos(-1.d0)

      write(*,*) 'give the timestep tau'
      read (*,*) tau
      write(*,*) 'mu1 and mu2 in electron mass units'
      read (*,*) mu1,mu2
      write(*,*) 'Z1 and Z2'
      read (*,*) z1,z2
      write(*,*) '# timestep iterations, n >= 1, n <=',mni
      read (*,*) ni
      ti =(ni+1)*tau
      write(*,*) 'imaginary time t = (n+1)*tau =',ti
      write(*,*) '# MC steps'
      read (*,*) mnmc
      write(*,*) 'give the kind of Coulomb potential (in,one,ong)'
      read (*,*) potk
      write(*,*) '....................................................'
      write(*,*) 'Sphere radius       a =',a
      write(*,*) 'time step         tau =',tau
      write(*,*) 'g distance step delta =',delta
      write(*,*) '....................................................'
      write(tabf,'(I2)') ni
      tabf=adjustl(tabf)
      tabf='table-'//trim(potk)//'-'//trim(tabf)//'.dat'
      open(unit=8,file=tabf,status='unknown')
      write(8,*) '#...................................................'
      write(8,*) '# Sphere radius       a =',a
      write(8,*) '# time step         tau =',tau
      write(8,*) '# mu1, mu2              =',mu1,mu2
      write(8,*) '# z1, z2                =',z1,z2
      write(8,*) '# ti                    =',ti
      write(8,*) '# kind of potential     =',potk
      write(8,*) '#...................................................'

      nd  = int(pi/delta)

c     loop over geodesic distance
      do id=1,nd
         th10=id*delta
         ph10=0.
         th20=0.
         ph20=0.
         ccc=cos(th10)*cos(th20)+sin(th10)*sin(th20)*cos(ph10-ph20)
         if(ccc.gt.1.)  ccc=1.
         if(ccc.lt.-1.) ccc=-1.
         dg0=a*acos(ccc)
         de0=2*a*sin(dg0/2/a)
         pcdm = 0.
         pidm = 0.
         pcdm2= 0.
         pidm2= 0.
c     MC step
         nmc  = 0
 10      nmc=nmc+1
 30      do k=1,ni
            ith1(k)=rand()*pi
            ith2(k)=rand()*pi
            iph1(k)=rand()*2*pi
            iph2(k)=rand()*2*pi
         enddo
         th1=ith1(1)
         ph1=iph1(1)
         th2=ith2(1)
         ph2=iph2(1)
         ccc=cos(th1)*cos(th2)+sin(th1)*sin(th2)*cos(ph1-ph2)
         if(ccc.gt.1.)  ccc=1.
         if(ccc.lt.-1.) ccc=-1.
         dg=a*acos(ccc)
         de=2*a*sin(dg/2/a)
         if(dg.lt.epsd)goto 30
         rhor=rho(potk,tau,a,mu1,mu2,z1,z2,
     &        th10,ph10,th20,ph20,th1,ph1,th2,ph2,dg,de,dg0,de0)
         rhoidr=rhoid(tau,a,mu1,mu2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20)
         if(rhor.eq.0..or.rhoidr.eq.0.)goto 30
         fa=rhor
         faid=rhoidr
         do j=2,ni
            th1=ith1(j-1)
            ph1=iph1(j-1)
            th2=ith2(j-1)
            ph2=iph2(j-1)
            th1p=ith1(j)
            ph1p=iph1(j)
            th2p=ith2(j)
            ph2p=iph2(j)
            ccc=cos(th1)*cos(th2)+sin(th1)*sin(th2)*cos(ph1-ph2)
            if(ccc.gt.1.)  ccc=1.
            if(ccc.lt.-1.) ccc=-1.
            dg=a*acos(ccc)
            de=2*a*sin(dg/2/a)
            ccc=cos(th1p)*cos(th2p)+sin(th1p)*sin(th2p)*cos(ph1p-ph2p)
            if(ccc.gt.1.)  ccc=1.
            if(ccc.lt.-1.) ccc=-1.
            dgp=a*acos(ccc)
            dep=2*a*sin(dgp/2/a)
            if(dg .lt.epsd)goto 30
            if(dgp.lt.epsd)goto 30
            rhor=rho(potk,tau,a,mu1,mu2,z1,z2,
     &           th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p,dg,de,dgp,dep)
            rhoidr=rhoid(tau,a,mu1,mu2,
     &           th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p)
            if(rhor.eq.0..or.rhoidr.eq.0.)goto 30
            fa=fa*rhor
            faid=faid*rhoidr
         enddo
         th1=ith1(ni)
         ph1=iph1(ni)
         th2=ith2(ni)
         ph2=iph2(ni)
         ccc=cos(th1)*cos(th2)+sin(th1)*sin(th2)*cos(ph1-ph2)
         if(ccc.gt.1.)  ccc=1.
         if(ccc.lt.-1.) ccc=-1.
         dg=a*acos(ccc)
         de=2*a*sin(dg/2/a)
         if(dg.lt.epsd)goto 30
         rhor=rho(potk,tau,a,mu1,mu2,z1,z2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20,dg,de,dg0,de0)
         rhoidr=rhoid(tau,a,mu1,mu2,
     &        th1,ph1,th2,ph2,th10,ph10,th20,ph20)
         if(rhor.eq.0..or.rhoidr.eq.0.)goto 30
         fa=fa*rhor
         faid=faid*rhoidr
         pcdm   = pcdm+fa
         pidm   = pidm+faid
         pcdm2  = pcdm2+fa**2.
         pidm2  = pidm2+faid**2.
         avpcdm = pcdm/nmc
         avpidm = pidm/nmc
         avpcdm2= pcdm2/nmc
         avpidm2= pidm2/nmc
         sigmac = sqrt((avpcdm2-avpcdm**2.)/nmc)/avpcdm
         sigmai = sqrt((avpidm2-avpidm**2.)/nmc)/avpidm
         if(mod(nmc,mnmc).eq.0)then
            print *,'..................Rel. Err....................'
            print *,nmc,sigmac,sigmai
            print *,'..............................................'
         endif
c         if(nmc.gt.mnmc)goto 20
         if(nmc.gt.mnmc.and.sigmac.lt.epsmc.and.sigmai.lt.epsmc)goto 20
         goto 10
c     tabulate P=-log(rho/rho0)
 20      pcdm = -log(avpcdm/avpidm)
         write(8,100) dg0,pcdm,ti*pot(potk,z1,z2,dg0,de0),
     &        pcdm/(ti*pot(potk,z1,z2,dg0,de0))
         call flush(8)
         write(*,*)   dg0,pcdm,ti*pot(potk,z1,z2,dg0,de0),
     &        pcdm/(ti*pot(potk,z1,z2,dg0,de0))
      enddo

 100  format(4(d10.3,3x))
      close(unit=8)

      stop
      end


      function fact(n)
      implicit none
c     factorial function
      real*8    fact,p
      integer   i,n

      p=1.
      do i=1,n
         p=p*dble(i)
      enddo
      fact=p

      return
      end


      function vr(a,mu,th)
      implicit none
c     Eq. (2.32) Bastianelli and Corradini EPJC 77, 731 (2017)
c     choose the Riemann normal coordinatees with origin on the
c     north pole th=0
      real*8    vr,a,mu,th
      real*8    m,mx,eps
      eps=1.d-6

      m =1/a
      mx=th
      if(mx.eq.0.) mx=eps
      vr=-m**2./6-(5*mx**2.-3+(mx**2.+3)*cos(2*mx))/48/(mx*sin(mx)/m)**2.
c     cutoff at the south pole
      if(vr.lt.-10.) vr=-10.

      return
```

```
      end


      function pot(potk,z1,z2,dg,de)
      implicit none
c     the Coulomb potential
      real*8    pot,z1,z2
      real*8    de,dg
      character potk*3

      if(potk.eq.'in') then
         pot=-z1*z2*log(de)
      elseif(potk.eq.'one') then
         pot=z1*z2/de
      elseif(potk.eq.'ong') then
         pot=z1*z2/dg
      else
         write(*,*) 'no pair-potential available !'
         stop
      endif

      return
      end


      function rho(potk,tau,a,mu1,mu2,z1,z2,
     &     th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p,dg,de,dgp,dep)
      implicit none
c     primitive approximation for rho
      real*8    rho,tau,a,mu1,mu2,z1,z2,dg,de,dgp,dep
      real*8    th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p
      real*8    pot,vr,ddx1,ddx2
      character potk*3

      ddx1=(a*th1*cos(ph1)-a*th1p*cos(ph1p))**2.+
     &     (a*th1*sin(ph1)-a*th1p*sin(ph1p))**2.
      ddx2=(a*th2*cos(ph2)-a*th2p*cos(ph2p))**2.+
     &     (a*th2*sin(ph2)-a*th2p*sin(ph2p))**2.

      rho =    exp(-mu1*ddx1/2/tau-tau*
     &     (vr(a,mu1,th1)+vr(a,mu1,th1p))/2)
      rho =rho*exp(-mu2*ddx2/2/tau-tau*
     &     (vr(a,mu2,th2)+vr(a,mu2,th2p))/2)
      rho =rho*exp(-tau*(pot(potk,z1,z2,dg,de)+
     &     pot(potk,z1,z2,dgp,dep))/2)

      return
      end


      function rhoid(tau,a,mu1,mu2,
     &     th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p)
      implicit none
c     primitive approximation for ideal gas rho
      real*8    rhoid,tau,a,mu1,mu2
      real*8    th1,ph1,th2,ph2,th1p,ph1p,th2p,ph2p
      real*8    vr,ddx1,ddx2

      ddx1=(a*th1*cos(ph1)-a*th1p*cos(ph1p))**2.+
     &     (a*th1*sin(ph1)-a*th1p*sin(ph1p))**2.
      ddx2=(a*th2*cos(ph2)-a*th2p*cos(ph2p))**2.+
     &     (a*th2*sin(ph2)-a*th2p*sin(ph2p))**2.

      rhoid =      exp(-mu1*ddx1/2/tau-tau*
     &     (vr(a,mu1,th1)+vr(a,mu1,th1p))/2)
      rhoid =rhoid*exp(-mu2*ddx2/2/tau-tau*
     &     (vr(a,mu2,th2)+vr(a,mu2,th2p))/2)

      return
      end


      function rho0(a,mu,t)
      implicit none
c     EQ. (3.23) Bastianelli and Corradini EPJC 77, 731 (2017)
      real*8    rho0,a,mu,t,th
      real*8    pi,fact,rit
      real*8    c1,c2,c3,c4,c5,c6,c7,c8
      parameter(c1=1./2,c2=16.,c3=59./4,c4=58./3,c5=550789./1260,
     &     c6=46838./45,c7=596004707./720)
      pi=acos(-1.d0)

      rit=t*2./a**2.
      rho0=rit/12+c1*rit**2./fact(6)+c2*rit**3./fact(9)+
     &     c3*rit**4./fact(10)+c4*rit**5./fact(11)+c5*rit**6./fact(13)+
     &     c6*rit**7./fact(14)+c7*rit**8./fact(17)
      rho0=mu*exp(rho0)/(2.*pi*t)

      return
      end
```

## AUTHOR DECLARATIONS

### Conflicts of interest

None declared.

### Data availability

The data that support the findings of this study are available from the corresponding author upon reasonable request.


### Funding

None declared.


---


[1] D. M. Ceperley, Path integrals in the theory of condensed Helium, Rev. Mod. Phys. **67**, 279 (1995).
[2] R. G. Storer, Path-Integral Calculation of the Quantum-Statistical Density Matrix for Attractive Coulomb Forces, J. Math. Phys. **9**, 964 (1968).
[3] A. D. Klemm and R. G. Storer, The structure of quantum fluids: helium and neon, Aust. J. Phys. **26**, 43 (1973).
[4] H. F. Trotter, On the Product of Semi-Groups of Operators, Proc. Am. Math. Soc. **10**, 545 (1959).
[5] J. C. Kimball, Short-Range Correlations and Electron-Gas Response Functions, Phys. Rev. A **7**, 1648 (1973).
[6] E. L. Pollock, Properties and computation of the Coulomb pair density matrix, Comput. Phys. Commun. **52**, 49 (1988).
[7] I. H. Duru and H. Kleinert, Quantum Mechanics of H-Atom from Path Integrals, Fortschritte der Physik **30**, 401 (1982).
[8] P. Vieillefosse, Coulomb Pair Density Matrix I, J. Stat. Phys. **74**, 1195 (1994).
[9] P. Vieillefosse, Coulomb Pair Density Matrix II, J. Stat. Phys. **80**, 461 (1995).
[10] R. Fantoni, Plasma living in a curved surface at some special temperature, Physica A **177**, 524 (2019).
[11] F. Bastianelli and O. Corradini, On the simplified path integral on spheres, Eur. Phys. J. C **77**, 731 (2017).

[12] M. H. Kalos and P. A. Whitlock, *Monte Carlo Methods* (John Wiley & Sons Inc., New York, 1986).
[13] R. Fantoni, The density of a fluid on a curved surface, J. Stat. Mech. , P10024 (2012).